

# DISPLAY

**DISPLAY** [( *rep*)] [*options*] {[ / ... ] [*output-format*] *output-element*} ...

This chapter covers the following topics:

- Function
- Syntax Description
- Defaults Applicable for a DISPLAY Statement
- Examples

For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

Related Statements: AT END OF PAGE | AT TOP OF PAGE | CLOSE PRINTER | DEFINE PRINTER EJECT | FORMAT | NEWPAGE | PRINT | SKIP | SUSPEND IDENTICAL SUPPRESS | WRITE | WRITE TITLE | WRITE TRAILER

Belongs to Function Group: *Creation of Output Reports*

---

## Function

The DISPLAY statement is used to specify the fields to be output on a report in column format. A column is created for each field and a field header is placed over the column.

### Note:

The statements WRITE and PRINT can be used to produce output in free (non-column) format.

See also the following topics (in the *Programming Guide*):

- *Controlling Data Output*
- *Statements DISPLAY and WRITE*
- *Index Notation for Multiple-Value Fields and Periodic Groups*
- *Column Headers*
- *Layout of an Output Page*

## Syntax Description

Syntax Element	Description
<i>(rep)</i>	<p><b>Report Specification:</b></p> <p>The notation <i>(rep)</i> may be used to specify the identification of the report for which the DISPLAY statement is applicable.</p> <p>As report identification, a value in the range 0 - 31 or a logical name which has been assigned using the DEFINE PRINTER statement may be specified.</p> <p>If <i>(rep)</i> is not specified, the statement will apply to the first report (Report 0).</p> <p>If this printer file is defined to Natural as PC, the report will be downloaded to the PC, see <i>Example 8</i>.</p> <p>For information on how to control the format of an output report created with Natural, see <i>Controlling Data Output</i> in the <i>Programming Guide</i>.</p>
<i>options</i>	<p><b>Display Options:</b></p> <p>For details, see <i>Display Options</i> below.</p>
<i>output-format</i>	<p><b>Output Format Definitions:</b></p> <p>For details, see <i>Output Format Definitions</i> below.</p>
/	<p><b>Line Advance - Slash Notation:</b></p> <p>When specified within a text element, a slash (/) causes a line advance for the text displayed.</p> <p>When specified between output elements, it causes the output element specified by the slash (/) to be placed vertically within the same column. The header for this column will be constructed by placing the headers of the vertically displayed elements vertically above the column.</p> <p>See also the following topics in the <i>Programming Guide</i>:</p> <ul style="list-style-type: none"> <li>● <i>Line Advance - Slash Notation</i></li> <li>● <i>Example 1 - Line Advance in DISPLAY Statement</i></li> <li>● <i>Suppressing Column Headers - Slash Notation</i></li> </ul>
<i>output-element</i>	<p><b>Output Element:</b></p> <p>For details, see <i>Output Element</i> below.</p>

## Display Options

<p>[NOTITLE] [NOHDR] [ [AND] [GIVE] [SYSTEM] FUNCTIONS ] [(statement-parameters)]</p>
---

## Syntax Element Description:

Syntax Element	Description
NOTITLE	<p><b>Default Page Title Suppression:</b></p> <p>By default, Natural generates a single title line for each page resulting from a <code>DISPLAY</code> statement. This title contains the page number, the time of day, and the date. Time of day is set at the beginning of the program execution or at the beginning of the job (batch mode). The default title line may be overridden by using a <code>WRITE TITLE</code> statement, or it may be suppressed by specifying the keyword <code>NOTITLE</code> in the <code>DISPLAY</code> statement.</p> <p>Examples:</p> <ul style="list-style-type: none"><li>● Default title will be produced: <code>DISPLAY NAME</code></li><li>● User title will be produced: <code>DISPLAY NAME WRITE TITLE 'user-title'</code></li><li>● No title will be produced: <code>DISPLAY NOTITLE NAME</code></li></ul> <p><b>Note:</b> If the <code>NOTITLE</code> option is used, it applies to all <code>DISPLAY</code>, <code>PRINT</code> and <code>WRITE</code> statements within the same object which write data to the same report.</p>

Syntax Element	Description
NOHDR	<p><b>Column Headers:</b></p> <p>Column headers are produced for each field specified in the DISPLAY statement using the following rules:</p> <ul style="list-style-type: none"> <li>• The header text may be explicitly specified in the DISPLAY statement before the field name. For example:  <pre>DISPLAY 'EMPLOYEE' NAME 'SALARY' SALARY</pre> </li> <li>• If you do not specify an explicit header for a field, the header as defined in the DEFINE DATA statement will be used.</li> <li>• If for a database field no header is defined in the DEFINE DATA statement, the default header as defined in the DDM will be used.</li> <li>• If no default header is defined in the DDM, the field name will be used as header.</li> <li>• If for a user-defined variable no header is defined in the DEFINE DATA statement, the variable name will be used as header. See also the DEFINE DATA statement for header definition.  <pre>DISPLAY NAME SALARY #NEW-SALARY</pre> </li> <li>• Natural always underlines column headings and generates one blank line between the underlining and the data being displayed.</li> <li>• If there are multiple DISPLAY statements in a program, the first DISPLAY statement determines the column header(s) to be used; this is evaluated at compilation time.</li> </ul> <p><b>Column Header Suppression:</b></p> <p>To suppress the column header for a single field</p> <ul style="list-style-type: none"> <li>• Specify the following characters (apostrophe-slash-apostrophe) before the field name:  <pre>'/'</pre> </li> </ul> <p>For example:  <pre>DISPLAY '/' NAME 'SALARY' SALARY</pre> </p> <p><b>To suppress all column headers</b></p> <ul style="list-style-type: none"> <li>• Specify the keyword NOHDR:  <pre>DISPLAY NOHDR NAME SALARY</pre> </li> </ul> <p><b>Notes:</b></p> <ol style="list-style-type: none"> <li>1. NOHDR only takes effect for the first DISPLAY statement, as subsequent DISPLAY statements cannot create column headers anyhow.</li> <li>2. If both NOTITLE and NOHDR are used, they must be specified in the following order: DISPLAY NOTITLE NOHDR NAME SALARY</li> </ol>

Syntax Element	Description
GIVE SYSTEM FUNCTIONS	<p><b>Natural System Function Usage:</b></p> <p>The GIVE SYSTEM FUNCTIONS clause is used to make available the following Natural system functions: AVER, COUNT, MAX, MIN, NAVER, NCOUNT, NMIN, SUM, TOTAL. These are evaluated when the DISPLAY statement containing the GIVE SYSTEM FUNCTIONS clause is executed.</p> <p>These functions may then be referred to in a statement executed as a result of an end-of-page condition.</p> <p><b>Notes:</b></p> <ol style="list-style-type: none"> <li>1. Only one DISPLAY statement per report may contain a GIVE SYSTEM FUNCTIONS clause. When system functions are evaluated from a DISPLAY statement, they are evaluated on a page basis, which means that all functions (except TOTAL) are reset to zero when a new page is initiated.</li> <li>2. When system functions are used within a DISPLAY statement within a subroutine, the end-of-page processing must occur within the same routine.</li> </ol> <p>See also <i>Example 2 - DISPLAY Statement Using GIVE SYSTEM FUNCTIONS Clause</i>.</p>
<i>statement-parameters</i>	<p><b>Parameter Definition at Statement Level:</b></p> <p>One or more parameters, enclosed within parentheses, may be specified at statement level, that is, immediately after the DISPLAY statement.</p> <p>Each parameter specified will override the corresponding parameter previously specified in a GLOBALS command, SET GLOBALS (Reporting Mode only) or FORMAT statement.</p> <p>If more than one parameter is specified, they must be separated by one or more blanks from one another. Each parameter specification must not be split between two statement lines.</p> <p><b>Note:</b></p> <p>The parameter settings applied here will only be regarded for variable fields, but they have no effect on text-constants. If you would like to set field attributes for a text-constant, they have to be set explicitly for this element, see <i>Parameter Definition at Element (Field) Level</i>.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>List of Parameters</i></li> <li>● <i>Example of Parameter Usage at Statement and Element (Field) Level</i></li> <li>● <i>Example 7 - DISPLAY Statement Using Parameters on Statement/Element Level</i></li> </ul>

## List of Parameters

The following parameters can be specified with the DISPLAY statement

<b>Parameter Name</b>	<b>Explanation</b>	<b>Specification possible at statement level (S), at element level (E) or both (SE)</b>
AD	Attribute Definition	SE
AL	Alphanumeric Length for Output	SE
CD	Color Definition	SE
CV	Control Variable	SE
DF	Date Format	SE
DL	Display Length for Output	SE
DY	Dynamic Attributes	SE
EM	Edit Mask	SE
EMU	Unicode Edit Mask	E
ES	Empty Line Suppression	S
FC	Filler Character	SE
FL	Floating Point Mantissa Length	SE
GC	Filler Character for Group Headers	SE
HC	Header Centering	SE
HW	Heading Width	SE
IC	Insertion Character	SE
IS	Identical Suppress	SE
LC	Leading Characters	SE
LS	Line Size	S
MC	Multiple-Value Field Count	S
MP	Maximum Number of Pages of a Report	S
NL	Numeric Length for Output	SE
PC	Periodic Group Count	S
PM	Print Mode	SE
PS	Page Size	S
SF	Spacing Factor	SE
SG	Sign Position	SE
TC	Trailing Characters	SE
UC	Underlining Character	SE
ZP	Zero Printing	SE

The individual parameters are described in the *Parameter Reference* (session parameters).

See also the following topics in the *Programming Guide*:


- *Centering of Column Headers - HC Parameter*
- *Width of Column Headers - HW Parameter*
- *Filler Characters for Headers - Parameters FC and GC*
- *Underlining Character for Titles and Headers - UC Parameter*

### Example of Parameter Usage at Statement and Element (Field) Level

```

DEFINE DATA LOCAL
1 VARI (A4)          INIT <'1234'>          /*      Output
END-DEFINE          /*      Produced
*                  /*      -----
DISPLAY NOHDR      'Text'                  '='    VARI          /*      Text 1234
DISPLAY NOHDR (AD=U) 'Text'                  '='    VARI          /*      Text 1234
DISPLAY NOHDR      'Text' (AD=U)           '='    VARI (AD=U) /*      Text 1234
DISPLAY NOHDR      'Text' (AD=U)           '='    VARI          /*      Text 1234
END
    
```

### Output Format Definitions



### Field Positioning Notations

Syntax Element	Description
$nX$	<p><b>Column Spacing:</b></p> <p>This notation inserts <math>n</math> spaces between columns.</p> <p>Example:</p> <pre>DISPLAY NAME 5X SALARY</pre> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 1 - DISPLAY Statement Using <math>nX</math> and <math>nT</math> Notation</i> (below)</li> <li>● <i>Column Spacing - SF Parameter and <math>nX</math> Notation</i> (in the <i>Programming Guide</i>)</li> </ul>
$nT$	<p><b>Tab Setting:</b></p> <p>The <math>nT</math> notation causes positioning (tabulation) to display position <math>n</math>. Backward positioning is not permitted.</p> <p>In the following example, NAME is displayed beginning in position 25, and SALARY beginning in position 50:</p> <pre>DISPLAY 25T NAME 50T SALARY</pre> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 1 - DISPLAY Statement Using <math>nX</math> and <math>nT</math> Notation</i> (below)</li> <li>● <i>Tab Setting - <math>nT</math> Notation</i> (in the <i>Programming Guide</i>)</li> </ul>
$x/y$	<p><b><math>x/y</math> Positioning:</b></p> <p>The <math>x/y</math> notation causes the next element to be placed <math>x</math> lines below the output of the last statement, beginning in column <math>y</math>. <math>y</math> must not be zero. Backward positioning is not permitted.</p>
$T^*field-name$	<p><b>Field Related Positioning:</b></p> <p>The <math>T^*</math> notation is used to position to a specific print position of a field used in a previous DISPLAY statement. Backward positioning is not permitted.</p>
$P^*field-name$	<p><b>Field and Line Related Positioning:</b></p> <p>The <math>P^*</math> notation is used to position to a specific print position and line of a field used in a previous DISPLAY statement. It is most often used in conjunction with vertical display mode. Backward positioning is not permitted.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Example 3 - DISPLAY Statement Using <math>P^*</math> Notation</i> (below)</li> <li>● <i>Tab Notation <math>P^*field</math></i> (in the <i>Programming Guide</i>)</li> </ul>




## Override Column Heading Assignment

Syntax Element	Description
'text'  '/'	<p><b>Text Assignment:</b></p> <p>If placed immediately before a field, the text enclosed by single quotes overrides the column heading.</p> <p>The slash character '/' before a field causes the header for the field to be suppressed.</p> <pre>DISPLAY 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT</pre> <p>If multiple 'text' elements are specified before a field name, the <i>last</i> 'text' element will be used as the column header and the other text elements will be placed before the value of the field within the column.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Define Your Own Column Headers (in the Programming Guide)</i></li> <li>● <i>Text Notation, Defining a Text to Be Used with a Statement (in the Programming Guide)</i></li> <li>● <i>Example 4 - DISPLAY Statement Using 'text', 'c(n)' and Attribute Notation (below)</i></li> </ul>
'c'(n)	<p><b>Character Repetition:</b></p> <p>The character enclosed by single quotes is displayed <i>n</i> times immediately before the field value. For example:</p> <pre>DISPLAY '*' (5) '=' NAME</pre> <p>results in</p> <pre>***** SMITH</pre> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Text Notation, Defining a Character to Be Displayed n Times before a Field Value (in the Programming Guide)</i></li> <li>● <i>Example 4 - DISPLAY Statement Using 'text', 'c(n)' and Attribute Notation (below)</i></li> </ul>

## Output Attributes

*attributes* indicates the output attributes to be used for text display. Attributes may be:



For the possible session parameter values, refer to the corresponding sections in the *Parameter Reference* documentation:

- *AD - Attribute Definition*, section *Field Representation*
- *CD - Color Definition*
- *PM - Print Mode*

**Note:**

The compiler actually accepts more than one attribute value for an output field. For example, you may specify: AD=BDI. In such a case, however, only the last value applies. In the given example, only the value I will become effective and the output field will be displayed intensified.

**Vertical/Horizontal Display**

The VERT clause may be used to cause multiple field values to be positioned underneath one another in the same column. In vertical mode, a new column may be initiated by specifying the keyword VERT or HORIZ.

The column heading in vertical mode is controlled using the entry or entries specified with the AS clause as described below.

Syntax Element	Description
<u>VERTICALLY</u>	<p><b>DISPLAY VERT without AS Clause:</b> Vertical column orientation. No column heading is produced if the AS clause is omitted.</p> <p>DISPLAY VERT NAME SALARY</p> <p>For an example, see <i>DISPLAY VERT without AS Clause</i> in the <i>Programming Guide</i>.</p>
AS 'text'	<p><b>DISPLAY VERT AS 'text' Clause:</b> Vertical column orientation. If AS 'text' is specified, the text enclosed by single quotes is used as the column heading.</p> <p>For an example, see <i>DISPLAY VERT AS 'text'</i> in the <i>Programming Guide</i>.</p> <p>The slash character / in the character string of 'text' will cause multiple lines of column headings.</p> <p>DISPLAY VERT AS 'LAST/NAME' NAME</p>
AS 'text' CAPTIONED	<p><b>DISPLAY VERT AS 'text' CAPTIONED Clause:</b> Vertical column orientation. If AS 'text' CAPTIONED is specified, 'text' is used as the column heading and the standard heading text or field name is inserted immediately before the field value in each detail display line.</p> <p>DISPLAY VERT AS 'PERSONS/SELECTED' CAPTIONED NAME FIRST-NAME</p> <p>For an example, see <i>DISPLAY VERT AS 'text' CAPTIONED</i> in the <i>Programming Guide</i>.</p>
AS CAPTIONED	<p><b>DISPLAY VERT AS CAPTIONED Clause:</b> Vertical column orientation. If AS CAPTIONED is specified, the standard heading text for the field (either heading text or the field name) will be used as the column heading.</p> <p>DISPLAY VERT AS CAPTIONED NAME FIRST-NAME</p>
<u>HORIZONTALLY</u>	<p><b>DISPLAY HORIZ Clause:</b> Horizontal column orientation. This is the default display mode.</p>

Vertical and horizontal column orientation may be intermixed by using the respective keyword.

To suspend vertical display for a single output element, you may place a dash (-) in front of the element. For example:

```
DISPLAY VERT NAME - FIRST-NAME SALARY
```

In the above example, FIRST-NAME will be output horizontally next to NAME, while SALARY will be output vertically again, i.e. below NAME.

The standard display mode is horizontal. A column is constructed for each field to be displayed.

Column headings are obtained and used by Natural according to the following priority:

1. heading 'text' supplied in the DISPLAY statement;
2. the default heading defined in the DDM (database fields), or the name of a user-defined variable;
3. the field name as defined in the DDM (if no heading text was defined for the database field).

For group names, a group heading is produced for the entire group. When specifying a group, only the heading for the entire group may be overridden by a user-specified heading.

The maximum number of column header lines is 15.

Line size overflow is not permitted for output resulting from a DISPLAY statement. If a line overflow occurs, an error message is issued.

For more information about vertical/horizontal display usage, see:

- *Example 5 - DISPLAY Statement Using Horizontal Display*
- *Example 6 - DISPLAY Statement Using Vertical and Horizontal Display*
- *DISPLAY VERT AS CAPTIONED and HORIZ (in the Programming Guide)*

## Output Element

```

[ { 'text' [(attributes)] }
  { 'c'(n) [(attributes)] }...
  nX
  nT
  x/y ] ['='] {operand1 [(parameters)]}
    
```

Operand Definition Table:

Operand	Possible Structure	Possible Formats	Referencing Permitted	Dynamic Definition
<i>operand1</i>	S A G N	A N P I F B D T L G O	yes	no

Syntax Element Description

Syntax Element	Description
<i>nX</i>	<b>Column Spacing:</b> This is the same as under <i>Output Format Definitions</i> (see above).
<i>nT</i>	<b>Tab Setting:</b> This is the same as under <i>Output Format Definitions</i> (see above).
<i>x/y</i>	<b>x/y Positioning:</b> This is the same as under <i>Output Format Definitions</i> (see above).
' <i>text</i> '	<b>Text Assignment:</b> This is the same as under <i>Output Format Definitions</i> (see above).
' <i>c</i> ' ( <i>n</i> )	<b>Character Repetition:</b> This is the same as under <i>Output Format Definitions</i> (see above).
' <i>text</i> ' '=' ' <i>c</i> ' ( <i>n</i> ) '='	If ' <i>text</i> ' '=' is placed immediately before the field, <i>text</i> is output immediately before the field value. This applies analogously with ' <i>c</i> ' ( <i>n</i> ) '='.  DISPLAY '*****' '=' NAME
<i>attributes</i>	<b>Output Attributes:</b> This is the same as under <i>Output Format Definitions</i> (see above).
<i>operand1</i>	The field to be displayed.
<i>parameters</i>	<b>Parameter Definition at Element (Field) Level:</b>  One or more parameters, enclosed within parentheses, may be specified at element (field) level, that is, immediately after <i>operand1</i> . Each parameter specified in this manner will override the corresponding parameter previously specified at statement level or in a GLOBALS command, SET GLOBALS (in Reporting Mode only) or FORMAT statement.  If more than one parameter is specified, one or more blanks must be placed between each entry. An entry must not be split between two statement lines.  See also: <ul style="list-style-type: none"><li>● <i>List of Parameters</i></li><li>● <i>Example of Parameter Usage at Statement and Element (Field) Level</i></li></ul>

## Defaults Applicable for a DISPLAY Statement

The following defaults are applicable for a DISPLAY statement:

- **Report Width**

The width of the report defaults to the value set when Natural is installed. This default value is normally 132 in batch mode or the line length of the terminal in TP mode. It may be overridden with the session parameter `LS`. In TP mode, line size (`LS`) and page size (`PS`) parameters are set by Natural based on the physical characteristics of the terminal type in use.

- **Terminal Screen Output**

When the `DISPLAY` output is displayed on a terminal (emulation) screen, the output begins in physical Column 2 (because Column 1 must be reserved for possible use as an attribute position on a 3270-type terminal).

- **Printout on Paper**

When the `DISPLAY` output is printed on paper, the printout begins in the leftmost column (Column 1).

- **Spacing Factor**

The default spacing factor between elements is one position. There is a minimum of one space between columns (reserved for terminal attributes). This default may be overridden with the session parameter `SF`.

- **Field Output**

The length of the field or the field heading, whichever is greater, determines the column width for the report (unless the `HW` parameter is used).

- If the field is longer than the heading, the heading will be centered over the column unless the `HC=L` or `HC=R` parameter is used to produce a left-justified or right-justified heading.
- If the heading is longer than the field, the field will be left-justified under the heading.
- The values contained in the field are left-justified for alphanumeric fields and right-justified for numeric fields.
- Numeric fields may be displayed left-justified by specifying `AD=L`.
- Alphanumeric fields may be displayed right-justified by specifying `AD=R`.
- In a vertical display, the longest data value or heading among all fields determines the column width (unless the `HW` parameter is used).

- **Sign**

One extra high-order print position is reserved for a sign when printing a numeric field. The session parameter `SG` may be used to suppress the sign position.

- **Page Overflow**

Page overflow is checked before execution of a `DISPLAY` statement. No new page title or trailer information is generated during the execution of a `DISPLAY` statement.

## Examples

- Example 1 - `DISPLAY` Statement Using `nX` and `nT` Notation

- Example 2 - DISPLAY Statement Using GIVE SYSTEM FUNCTIONS Clause
- Example 3 - DISPLAY Statement Using P\* Notation
- Example 4 - DISPLAY Statement Using 'text ', 'c(n)' and Attribute Notation
- Example 5 - DISPLAY Statement Using Horizontal Display
- Example 6 - DISPLAY Statement Using Vertical and Horizontal Display
- Example 7 - DISPLAY Statement Using Parameters on Statement/Element Level
- Example 8 - Report Specification with Output File Defined to Natural as PC

**Example 1 - DISPLAY Statement Using nX and nT Notation**

```

** Example 'DISEX1': DISPLAY (with nX, nT notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE 5X NAME 50T JOB-TITLE
END-READ
*
END
    
```

**Output of Program DISEX1:**

NAME	CURRENT POSITION
ABELLAN	MAQUINISTA
ACHIESON	DATA BASE ADMINISTRATOR
ADAM	CHEF DE SERVICE
ADKINSON	PROGRAMMER

**Example 2 - DISPLAY Statement Using GIVE SYSTEM FUNCTIONS Clause**

```

** Example 'DISEX2': DISPLAY (with GIVE SYSTEM FUNCTIONS)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
*
LIMIT 15
FORMAT PS=15
*
READ EMPLOY-VIEW
  DISPLAY GIVE SYSTEM FUNCTIONS
    
```

```

PERSONNEL-ID NAME FIRST-NAME SALARY (1) CURR-CODE (1)
AT END OF PAGE
WRITE / 'SALARY STATISTICS:'
/ 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
/ 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
/ 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)
END-ENDPAGE
END-READ
*
END

```

**Output of Program DISEX2:**

Page 1 05-01-12 09:47:48

PERSONNEL ID	NAME	FIRST-NAME	ANNUAL SALARY	CURRENCY CODE
50005500	BLOND	ALEXANDRE	172000	FRA
50005300	MAIZIERE	ELISABETH	166900	FRA
50004900	CAUDAL	ALBERT	167350	FRA
50004600	VERDIE	BERNARD	170100	FRA
50004200	VAUZELLE	BERNARD	159790	FRA
50004100	CHAPUIS	ROBERT	169900	FRA
50003800	JOUSSELIN	DANIEL	171990	FRA
50006900	BAILLET	PATRICK	188000	FRA
50007600	MARX	JEAN-MARIE	365700	FRA

```

SALARY STATISTICS:
MAXIMUM: 365700 FRA
MINIMUM: 159790 FRA
AVERAGE: 192414 FRA

```

**Example 3 - DISPLAY Statement Using P\* Notation**

```

** Example 'DISEX3': DISPLAY (with P* notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
2 NAME
2 SALARY (1)
2 BIRTH
2 CITY
END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY CITY FROM 'N'
DISPLAY NOTITLE NAME CITY
VERT AS 'BIRTH/SALARY' BIRTH (EM=YYYY-MM-DD) SALARY (1)
SKIP 1
AT BREAK OF CITY
DISPLAY P*SALARY (1) AVER(SALARY (1))
SKIP 1
END-BREAK
END-READ
END

```



**Output of Program DISEX3:**

NAME	CITY	BIRTH SALARY
WILCOX	NASHVILLE	1970-01-01 38000
MORRISON	NASHVILLE	1949-07-10 36000
		37000

**Example 4 - DISPLAY Statement Using 'text ', 'c(n)' and Attribute Notation**

```

** Example 'DISEX4': DISPLAY (with 'c(n)' notation and attribute)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 DEPT
  2 LEAVE-DUE
  2 NAME
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY DEPT FROM 'T'
  IF LEAVE-DUE GT 40
    DISPLAY NOTITLE
      'EMPLOYEE' NAME           /* OVERRIDE STANDARD HEADER
      'LEAVE ACCUMULATED' LEAVE-DUE /* OVERRIDE STANDARD HEADER
      '**(10)(I)'              /* DISPLAY 10 '** INTENSIFIED
  ELSE
    DISPLAY NAME LEAVE-DUE
  END-IF
END-READ
*
END
    
```

**Output of Program DISEX4:**

EMPLOYEE	LEAVE ACCUMULATED
LAVENDA	33
BOYER	33
CORREARD	45
BOUVIER	19

**Example 5 - DISPLAY Statement Using Horizontal Display**

```

** Example 'DISEX5': DISPLAY (horizontal display)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
  2 SALARY (1:2)
  2 CURR-CODE (1:2)
END-DEFINE
    
```

```

*
LIMIT 4
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE NAME JOB-TITLE SALARY (1:2) CURR-CODE (1:2)
  SKIP 1
END-READ
*
END

```

**Output of Program DISEX5:**

NAME	CURRENT POSITION	ANNUAL SALARY	CURRENCY CODE
ABELLAN	MAQUINISTA	1450000 1392000	PTA PTA
ACHIESON	DATA BASE ADMINISTRATOR	11300 10500	UKL UKL
ADAM	CHEF DE SERVICE	159980 0	FRA 0
ADKINSON	PROGRAMMER	34500 31700	USD USD

**Example 6 - DISPLAY Statement Using Vertical and Horizontal Display**

```

** Example 'DISEX6': DISPLAY (vertical and horizontal display)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 JOB-TITLE
  2 SALARY (1:2)
  2 CURR-CODE (1:2)
END-DEFINE
*
LIMIT 1
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE VERT AS CAPTIONED
    NAME CITY 'POSITION' JOB-TITLE
    HORIZ 'SALARY' SALARY (1:2) 'CURRENCY' CURR-CODE (1:2)
  /*
  SKIP 1
END-READ
END

```

**Output of Program DISEX6:**

NAME CITY POSITION	SALARY	CURRENCY
ABELLAN	1450000	PTA
MADRID MAQUINISTA	1392000	PTA

### Example 7 - DISPLAY Statement Using Parameters on Statement/Element Level

```

** Example 'DISEX7': DISPLAY (with parameters for statement/element)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 PERSONNEL-ID
  2 TELEPHONE
  3 AREA-CODE
  3 PHONE
END-DEFINE
*
LIMIT 3
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE (AL=16 GC=+ NL=8 SF=3 UC==)
    PERSONNEL-ID NAME TELEPHONE (LC=< TC=>)
END-READ
END
    
```

#### Output of Program DISEX7:

PERSONNEL ID	NAME	AREA CODE	TELEPHONE
60008339	ABELLAN	<1 >	<4356726 >
30000231	ACHIESON	<0332 >	<523341 >
50005800	ADAM	<1033 >	<44864858 >

### Example 8 - Report Specification with Output File Defined to Natural as PC

```

** Example 'PCDIEX1': DISPLAY and WRITE to PC
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
  02 PERSONNEL-ID
  02 NAME
  02 CITY
END-DEFINE
*
FIND PERS WITH CITY = 'NEW YORK' /* Data selection
  WRITE (7) TITLE LEFT 'List of employees in New York' /
  DISPLAY (7) /* (7) designates the output file (here the PC).
    'Location' CITY
    'Surname' NAME
    'ID' PERSONNEL-ID
END-FIND
END
    
```